# #############################################################
# The pod browser application
# Tab spacing: 4
# No spaces in tabs
# #############################################################
PRECOMPILED_HEADER = precompiled.h
!include(../proto/proto.pri):error("Could not include the protocol buffer components")
!include(../GL/GL.pri):error("Could not include the OpenGL stuff")
HEADERS += mainWindow.h \
    pod.h \
    glDisplay.h
SOURCES += main.cpp \
    mainWindow.cpp \
    pod.cpp \
    glDisplay.cpp

# Set the forms
FORMS += ui/mainWindow.ui \
    ui/pod.ui
UI_DIR += ui
INCLUDEPATH += ..
RESOURCES += 
TARGET = podbrowser
CONFIG += qt \
    thread \
    debug_and_release \
    precompile_header
QT += opengl
TEMPLATE = app
DESTDIR = ../../bin

# This is needed to make sure stl:string works with QString without a weird conversion
DEFINES += QT_NO_CAST_TO_ASCII

# #############################################################
# Is this a static build?, if so make sure we get the image format plugins
static { 
    DEFINES += STATIC_BUILD
    QTPLUGIN += qjpeg \
        qgif \
        qtiff
}
